Delphi コードのクリーンアップ

# はじめに
業務アプリのコードを日々メンテしていると、度重なる仕様変更や機能の追加により、コンパイルするとヒントやワーニング (警告) が溜まり続ける事がありますよね？

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/21785/1f57df73-3fd4-3b06-96d6-1c111a51bca1.png)

正直にゆーてみ？再構築すると 3 桁とかあるじゃろ？

# コードのクリーンアップ
ヒントとワーニングをゼロにすると単純にコンパイル速度が上がり、ビルドに無駄な時間を割かなくて済みます。また、表面化していないバグを炙り出すのにも有効です。

仕方のない事ですが、古いアプリケーションをマイグレーションしようとすると (Delphi をバージョンアップすると) エラーと共に大量のヒントやワーニングが出る事があります。しかしながら、バグを仕込んでいる可能性も否定できないため、放置せずにキッチリと潰しておいた方がいいかと思います。

...でも、中には「どうやったらこれ消せるの？」ってヒントやワーニングがあったりしますよね？

## ■ ヒント
※ `$HINTS` 指令によってヒント表示を一括でオンオフする事が出来ます。
※ 個別にはオンオフできないので、個別に対応したい場合には `{$HINTS OFF}` `{$HINTS ON}` で括る必要があります。

**See also:**

 - [HINTS（Delphi）(DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/HINTS%EF%BC%88Delphi%EF%BC%89)

### ・H2077 '%s' に代入された値は使用されていません (Delphi)

**原因:**
メッセージの通り。

**対処方法:**
殆どのケースで、使われない代入は削除しても問題ない。

代入に何らかの計算式を伴っているのであれば、**計算結果が使われていない事がバグである可能性**もあるので、コメントアウトしてしばらく寝かせておくのもいいかもしれない。

このヒントを潰す退屈な作業をスクリプトを書いてサッサと終わらせたいと考えるかもしれない。コンパイルメッセージは IDE の [メッセージウィンドウ] のコンテキストメニューから保存できる。コンパイルログから `H2077` を含む行を探し、行番号を得て、当該行をコメントアウトしてしまえばよい...

```pascal
//  FValue := 0;
```

果たしてそうだろうか？

```pascal
//FValue := FuncA(10) +
            FuncB(5) * 120;
```

上記のようなコードは多くの場合でエラーになるのでまだいいが...

```pascal
  if a > 10 then
//  FValue := 0; 
  ProcA;
```

エラーもなくバグを生む事もある。

**See also:**

 - [H2077 '%s' に代入された値は使用されていません (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/H2077_%27%25s%27_%E3%81%AB%E4%BB%A3%E5%85%A5%E3%81%95%E3%82%8C%E3%81%9F%E5%80%A4%E3%81%AF%E4%BD%BF%E7%94%A8%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%9B%E3%82%93_%28Delphi%29)

### ・H2164 変数 '%s' が宣言されていますが '%s' の中では使用されていません (Delphi)

**原因:**
メッセージの通り。

**対処方法:**
殆どのケースで、使わない変数は削除しても問題ない。

**See also:**

 - [H2164 変数 '%s' が宣言されていますが '%s' の中では使用されていません (Delphi)(DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/H2164_%E5%A4%89%E6%95%B0_%27%25s%27_%E3%81%8C%E5%AE%A3%E8%A8%80%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%99%E3%81%8C_%27%25s%27_%E3%81%AE%E4%B8%AD%E3%81%A7%E3%81%AF%E4%BD%BF%E7%94%A8%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%9B%E3%82%93_%28Delphi%29)

### ・H2219 %s：private 部で宣言されていますが，クラス内でまったく使用されていません (Delphi)
**原因:**
メッセージの通り。

**対処方法1:**
このヒントを出しているフィールドやメソッドを将来使う可能性があるのであれば `{$HINTS OFF}` `{$HINTS ON}` で括り、ヒントを握りつぶす。

```pascal
  {$HINTS OFF}
  procedure HogeHoge;
  {$HINTS ON}
```

**対処方法2:**
もう使う事がないと判っているのであれば、素直にコードを削除した方がいい。

**See also:**

 - [H2219 %s：private 部で宣言されていますが，クラス内でまったく使用されていません (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/H2219_%25s%EF%BC%9Aprivate_%E9%83%A8%E3%81%A7%E5%AE%A3%E8%A8%80%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%99%E3%81%8C%EF%BC%8C%E3%82%AF%E3%83%A9%E3%82%B9%E5%86%85%E3%81%A7%E3%81%BE%E3%81%A3%E3%81%9F%E3%81%8F%E4%BD%BF%E7%94%A8%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%9B%E3%82%93_(Delphi))

## ■ ワーニング
※ `$WARN` 指令によってワーニング表示を個別にオンオフする事が出来ます。
※ ワーニングをエラーに昇格させる事もできます。

**See also:**

 - [警告メッセージ（Delphi）(DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/%E8%AD%A6%E5%91%8A%E3%83%A1%E3%83%83%E3%82%BB%E3%83%BC%E3%82%B8%EF%BC%88Delphi%EF%BC%89)


### ・W1000 シンボル '%s' を使用することは推奨されていません (Delphi)

**原因:**
互換性のために残されている機能を使っている (**deprecated** ヒント指令でマークされている機能を使っている)。

**対処方法 1:**
多くの場合 **deprecated** 指令のあるメソッドにはオーバーロードされたメソッドが存在するのでそちらを使う。そうでない場合にはドキュメントやコメントで代替方法が提示されているハズ。

**対処方法 2:**
プロジェクトのオプションにある `ヒントと警告` で `使用を推奨されていないシンボル` を **False** に設定する (Delphi のバージョンによってはチェックを外す)。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/21785/2d4cbff3-bb72-fb5a-3b63-90e4dd670a18.png)

非推奨には理由があるハズなので<font color="Crimson">この方法を採る事はオススメしない</font>。

**対処方法 3:**
当該箇所を `{$WARN SYMBOL_DEPRECATED OFF}` と `{$WARN SYMBOL_DEPRECATED ON}` で括ってもワーニングを消せるが、非推奨には理由があるハズなので<font color="Crimson">この方法を採る事はオススメしない</font>。

#### FileAge()

例えば次のような古いコードは `FileAge()` が非推奨の書式を使っているためワーニングが出る。

```pascal
var
  FileName: String;
  dt: TDateTime;
begin
  Filename := 'TEST.TXT';
  dt := FileDateToDateTime(FileAge(FileName));
  ...

```

そこで、非推奨でない書式の方の `FileAge()` をラップした次のような関数を用意して、

```pascal
function FileAgeToDateTime(const FileName: string): TDateTime;
begin
  if not FileAge(FileName, Result) then
    Result := 0;
end;
```

このように書き換える。

```pascal
var
  FileName: String;
  dt: TDateTime;
begin
  Filename := 'TEST.TXT';
  dt := FileAgeToDateTime(FileName);
  ...
end;  
```

#### Printer.GetPrinter()

12 Athens 以降だと `GetPrinter()` にオーバーロードされたメソッドが存在する。

```pascal:Vcl.Printers.pas
procedure GetPrinter(ADevice, ADriver, APort: PChar; var ADeviceMode: THandle); overload; deprecated;
procedure GetPrinter(var ADevice, ADriver, APort: string; var ADeviceMode: THandle); overload;
```

パラメータが **string** になっている方を使うように修正する。


**See also:**

 - [W1000 シンボル '%s' を使用することは推奨されていません (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1000_%E3%82%B7%E3%83%B3%E3%83%9C%E3%83%AB_%27%25s%27_%E3%82%92%E4%BD%BF%E7%94%A8%E3%81%99%E3%82%8B%E3%81%93%E3%81%A8%E3%81%AF%E6%8E%A8%E5%A5%A8%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%9B%E3%82%93_%28Delphi%29)
 - [System.SysUtils.FileAge (DocWiki)](http://docwiki.embarcadero.com/Libraries/ja/System.SysUtils.FileAge)

### ・W1002 シンボル '%s' はプラットフォームに依存すると宣言されています (Delphi)
**原因:**
特定のプラットフォームでしか使用できないものとしてマークされている機能を使用している。

**対処方法 1:**
プロジェクトのオプションにある `ヒントと警告` で `プラットフォーム依存のシンボル` を **False** に設定する (Delphi のバージョンによってはチェックを外す)。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/21785/f8d2296e-aa31-2f7c-28c3-b74312240816.png)

**対処方法 2:**
このワーニングを出している機能を `{$WARN SYMBOL_PLATFORM OFF}` `{$WARN SYMBOL_PLATFORM ON}` で括る。

```pascal
{$WARN SYMBOL_PLATFORM OFF}
  if DebugHook <> 0 then
    ReportMemoryLeaksOnShutdown := True;
{$WARN SYMBOL_PLATFORM ON}
```

**See also:**

 - [W1002 シンボル '%s' はプラットフォームに依存すると宣言されています (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1002_%E3%82%B7%E3%83%B3%E3%83%9C%E3%83%AB_%27%25s%27_%E3%81%AF%E3%83%97%E3%83%A9%E3%83%83%E3%83%88%E3%83%95%E3%82%A9%E3%83%BC%E3%83%A0%E3%81%AB%E4%BE%9D%E5%AD%98%E3%81%99%E3%82%8B%E3%81%A8%E5%AE%A3%E8%A8%80%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%99_(Delphi))

### ・W1005 ユニット '%s' はプラットフォームに依存すると宣言されています (Delphi)
**原因:**
特定のプラットフォームでしか使用できないユニットを使用している。

**対処方法 1:**
プロジェクトのオプションにある `ヒントと警告` で `プラットフォーム依存のユニット` を **False** に設定する (Delphi のバージョンによってはチェックを外す)。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/21785/b1eda1b1-34a5-8e0b-25f2-9127c79cf4ed.png)

**対処方法 2:**
このワーニングを出しているユニットを **uses** 句で指定しているユニット (またはプロジェクトファイル) の先頭に `{$WARN UNIT_PLATFORM OFF}` を追加する。

```pascal
unit Foo;
{$WARN UNIT_PLATFORM OFF}
...
```

**See also:**

 - [W1005 ユニット '%s' はプラットフォームに依存すると宣言されています (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1005_%E3%83%A6%E3%83%8B%E3%83%83%E3%83%88_%27%25s%27_%E3%81%AF%E3%83%97%E3%83%A9%E3%83%83%E3%83%88%E3%83%95%E3%82%A9%E3%83%BC%E3%83%A0%E3%81%AB%E4%BE%9D%E5%AD%98%E3%81%99%E3%82%8B%E3%81%A8%E5%AE%A3%E8%A8%80%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%99_(Delphi))

### ・W1006 ユニット '%s' を使用することは推奨されていません (Delphi)

**原因:**
互換性のために残されているユニットを使っている (**deprecated** ヒント指令でマークされているユニットを使っている)。

**対処方法 1:**
多くは代替方法が提示されているハズ。ユニットのコメントヘッダ等も確認してみる。

**対処方法 2:**
プロジェクトのオプションにある `ヒントと警告` で `使用を推奨されていないユニット` を **False** に設定する (Delphi のバージョンによってはチェックを外す)。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/21785/ede1c9de-961e-9262-27bd-a327e78c539e.png)

非推奨には理由があるハズなので<font color="Crimson">この方法を採る事はオススメしない</font>。

**対処方法 3:**
このワーニングを出しているユニットを **uses** 句で指定しているユニット (またはプロジェクトファイル) の先頭に `{$WARN UNIT_DEPRECATED OFF}` を追加してもワーニングを消せるが、非推奨には理由があるハズなので<font color="Crimson">この方法を採る事はオススメしない</font>。

#### EOleException
例えば、古いコードだと `EOleException` を使うのに `OleAuto` を **uses** してあると思うが、現在では `OleAuto` は非推奨としてマークされている。

```pascal
{*******************************************************}
{                                                       }
{            Delphi Visual Component Library            }
{                                                       }
{ Copyright(c) 1995-2021 Embarcadero Technologies, Inc. }
{              All rights reserved                      }
{                                                       }
{*******************************************************}

unit Vcl.OleAuto deprecated;

{$HPPEMIT LEGACYHPP}
{$DENYPACKAGEUNIT}

{ OleAuto cannot be used in a package DLL.  To implement
  an OLE automation server in a package, use the new
  OLE automation support in comobj and comserv.
}
```

コメントにあるように、現在では `EOleException` は `ComObj` で定義されているため、**uses** 句にある `OleAuto` を `ComObj` で置き換えるとワーニングを解消できる。

**See also:**

 - [W1006 ユニット '%s' を使用することは推奨されていません (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/Sydney/ja/W1006_%E3%83%A6%E3%83%8B%E3%83%83%E3%83%88_%27%25s%27_%E3%82%92%E4%BD%BF%E7%94%A8%E3%81%99%E3%82%8B%E3%81%93%E3%81%A8%E3%81%AF%E6%8E%A8%E5%A5%A8%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%9B%E3%82%93_%28Delphi%29)

### ・W1009 '%s' の再宣言のため基本クラスのメンバーが隠蔽されました (Delphi)
**原因:**
コンポーネントやフォームを派生した際、知らずに基底クラスのメンバーと同じ名前のプロパティを作ってしまったパターンが多いが、製品のバージョンアップに伴い、基底クラスに新しいメンバーが追加された事で発生する事もある。

`「意図しないプロパティ名競合の可能性」`

と読み替えるとスッキリするかと思う。

**対処方法1:**
リネームで解決するのならリネーム。

**対処方法2:**
RTTI を使っているなどの理由 (プロパティ名が外的要因でネーミングされている) により、プロパティ名をリネームできないのであれば、当該箇所を `{$WARN HIDING_MEMBER OFF}` と `{$WARN HIDING_MEMBER ON}` で括る。

```pascal
    {$WARN HIDING_MEMBER OFF}
    property UNITNAME: String read FUNITNAME write FUNITNAME;
    {$WARN HIDING_MEMBER ON}
```

**See also:**

 - [W1009 '%s' の再宣言のため基本クラスのメンバーが隠蔽されました (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1009_%27%25s%27_%E3%81%AE%E5%86%8D%E5%AE%A3%E8%A8%80%E3%81%AE%E3%81%9F%E3%82%81%E5%9F%BA%E6%9C%AC%E3%82%AF%E3%83%A9%E3%82%B9%E3%81%AE%E3%83%A1%E3%83%B3%E3%83%90%E3%83%BC%E3%81%8C%E9%9A%A0%E8%94%BD%E3%81%95%E3%82%8C%E3%81%BE%E3%81%97%E3%81%9F_(Delphi))


### ・W1010 メソッド '%s' で基底型 '%s' の仮想メソッドが隠蔽されます（Delphi）
**原因:**
コンポーネントやフォームを派生した際、知らずに基底クラスの仮想メソッドと同じ名前のメソッドを作ってしまったパターンが多いが、製品のバージョンアップに伴い、基底クラスに新しい仮想メソッドが追加された事で発生する事もある。

`「意図しないメソッド名競合の可能性」`

と読み替えるとスッキリするかと思う。

**対処方法:**
コンポーネントやフォームが目的の通りに動作しているのであれば、ワーニングの出ているメソッドに **reintroduce** キーワードを付けるのが手っ取り早い。

```pascal
procedure VirtuMethod; reintroduce;
```

基底クラスに同名の仮想メソッドがある事を知らずに同じ名前を付けたのであれば、気付いた時点でリネームして名前の競合を解消する。既にあちこちで使われてしまってリファクタリングやドキュメントの修正が大変になっているのなら、妥協して **reintroduce** キーワードを付けるしかない。

マイグレーション時に発生した (旧バージョンをビルドしても出ない) 際は、動作には影響がないので、多くの場合で **reintroduce** キーワードを付ければいいと思うが、旧バージョンでリネームして動作確認したものをマイグレーションするのが筋だとは思う。

**See also:**

 - [W1010 メソッド '%s' で基底型 '%s' の仮想メソッドが隠蔽されます（Delphi）(DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1010_%E3%83%A1%E3%82%BD%E3%83%83%E3%83%89_%27%25s%27_%E3%81%A7%E5%9F%BA%E5%BA%95%E5%9E%8B_%27%25s%27_%E3%81%AE%E4%BB%AE%E6%83%B3%E3%83%A1%E3%82%BD%E3%83%83%E3%83%89%E3%81%8C%E9%9A%A0%E8%94%BD%E3%81%95%E3%82%8C%E3%81%BE%E3%81%99%EF%BC%88Delphi%EF%BC%89)
 - [Reintroduce (再導入) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/%E3%83%A1%E3%82%BD%E3%83%83%E3%83%89%EF%BC%88Delphi%EF%BC%89#Reintroduce)


### ・W1011 コンパイラは 'end.' 以降の文字を無視します - (Delphi)
**原因:**
コンパイルエラーが出ていないのであれば、ユニット末尾の `end.` 以降にも何かが書かれている。

**対処方法:**
`end.` 以降をフリーに書けるメモ帳代わりに使いたいのなら、ユニットのどこかに `{$WARN GARBAGE OFF}` を記述しておくといい。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/21785/6d504dd1-e179-aa68-92e4-9a51ebc3b748.png)

プロジェクトのオプション (`.END 以降へのテキストの記述。コンパイラはこれを無視する。`) でもワーニングを消せるが、ユニットを他のプロジェクトで使いまわそうとすると、やっぱりそこでワーニングになってしまう。それと、あちこちのユニットの末尾をメモ帳代わりにするくらいなら、普通にテキストファイルを一つ用意してプロジェクトに追加した方がいいように思う。

**See also:**

 - [W1011 コンパイラは 'end.' 以降の文字を無視します - (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1011_%E3%82%B3%E3%83%B3%E3%83%91%E3%82%A4%E3%83%A9%E3%81%AF_%27end.%27_%E4%BB%A5%E9%99%8D%E3%81%AE%E6%96%87%E5%AD%97%E3%82%92%E7%84%A1%E8%A6%96%E3%81%97%E3%81%BE%E3%81%99_-_%28Delphi%29)


### ・W1020: クラス '%s' のインスタンスを作成していますが，このクラスは抽象メソッド '%s.%s' を含みます (Delphi)
**原因:**
抽象メソッドを持ったクラスのインスタンスを作成しようとしたか、基底クラスに抽象メソッドが追加された。

**対処方法:**
ドキュメントにあるように抽象メソッドを持ったクラスのインスタンスを作成しようとしたのかもしれないが、使っていたクラスの基底クラスに抽象メソッドが追加された時もこのワーニングが発生する。この場合、ワーニングの出たクラスに基底クラスで追加された抽象メソッドを実装しなくてはならない。

**See also:**

 - [x1020: クラス '%s' のインスタンスを作成していますが，このクラスは抽象メソッド '%s.%s' を含みます (Delphi) (DocWiki)](https://docwiki.embarcadero.com/RADStudio/ja/X1020_%E3%82%AF%E3%83%A9%E3%82%B9_%27%25s%27_%E3%81%AE%E3%82%A4%E3%83%B3%E3%82%B9%E3%82%BF%E3%83%B3%E3%82%B9%E3%82%92%E4%BD%9C%E6%88%90%E3%81%97%E3%81%A6%E3%81%84%E3%81%BE%E3%81%99%E3%81%8C%EF%BC%8C%E3%81%93%E3%81%AE%E3%82%AF%E3%83%A9%E3%82%B9%E3%81%AF%E6%8A%BD%E8%B1%A1%E3%83%A1%E3%82%BD%E3%83%83%E3%83%89_%27%25s.%25s%27_%E3%82%92%E5%90%AB%E3%81%BF%E3%81%BE%E3%81%99_(Delphi))


### ・W1030: コンパイラ指令に間違いがあります：'%s' (Delphi)
**原因:**
`W1030 間違ったコンパイル指令 : 'true'` というワーニングの場合、多くは古いプロジェクトを XE5 以降に持ってきた際に発生している。プロジェクトオプションの値が XE5 で変更になっている事が原因。

**対処方法:**
[プロジェクトオプション] でコンパイルオプションを開き、[ターゲット] で `すべての構成` を選んでから、デバッグ情報の値を **"デバッグ情報"** に変更。変更後、念のために他のターゲットも確認。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/21785/1d8d638c-4e81-4289-3b08-02b5a3f8edf2.png)
この際、<font color="Crimson">プロジェクトファイル (*.dpr) の中身が壊れる事がある</font>ので、事前に外部のテキストエディタ等へ内容をコピーしておく。必要なのはプロジェクトファイルの上部に生成されるコンパイル指令だけなので、その下はすべて退避させた内容を書き戻しても大丈夫。

`W1030 間違ったコンパイル指令 : 'xxxxx'` の xxxxx 部分でコンパイル指令が特定できない場合には、プロジェクトファイル (*.dpr, *.dproj) を削除して作り直した方がいい (早い) かもしれない。

**See also:**

 - [x1030: コンパイラ指令に間違いがあります：'%s' (Delphi) (DocWiki)](https://docwiki.embarcadero.com/RADStudio/ja/X1030_%E3%82%B3%E3%83%B3%E3%83%91%E3%82%A4%E3%83%A9%E6%8C%87%E4%BB%A4%E3%81%AB%E9%96%93%E9%81%95%E3%81%84%E3%81%8C%E3%81%82%E3%82%8A%E3%81%BE%E3%81%99%EF%BC%9A%27%25s%27_(Delphi))
 - [プロジェクトコンパイル時に「[警告] W1030 間違ったコンパイル指令 : 'true'」が出力される (SupportKB)](https://support.embarcadero.com/article/43628)

### ・W1036 変数 '%s' は初期化されていない可能性があります (Delphi)
**原因:**
変数に値が割り当てられていないまま、その値を使おうとした場合に発生する。

**対処方法:**
多くの場合、

```pascal
procedure Proc;
var
  a, b: Integer;
begin
  a := Random(10);
  if a < 5 then
    b := 10;
  Writeln(b); // ここでワーニング
end;
```

ブロックの先頭で変数を初期化するだけでいい。

```pascal
procedure Proc;
var
  a, b: Integer;
begin
  a := Random(10);
  b := 0; // 変数 b を初期化
  if a < 5 then
    b := 10;
  Writeln(b);
end;
```

**Delphi 10.3** 以降の**インライン変数宣言**を使えば割とスッキリ解決できる。

```pascal
procedure Proc;
begin
  var a := Random(10);
  var b := 0; // 変数 b を初期化
  if a < 5 then
    b := 10;
  Writeln(b);
end;
```

この対処を行うと、新しいバージョンでプロジェクトを開いた時、「[H2077 '%s' に代入された値は使用されていません](#h2077-s-%E3%81%AB%E4%BB%A3%E5%85%A5%E3%81%95%E3%82%8C%E3%81%9F%E5%80%A4%E3%81%AF%E4%BD%BF%E7%94%A8%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%BE%E3%81%9B%E3%82%93-delphi)」 が出るようになる事がある (どうしろと...)。こればかりは `{$IF CompilerVersion ～}` コンパイラ指令で回避するしかないかもしれない。

**See also:**

 - [W1036 変数 '%s' は初期化されていない可能性があります (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1036_%E5%A4%89%E6%95%B0_%27%25s%27_%E3%81%AF%E5%88%9D%E6%9C%9F%E5%8C%96%E3%81%95%E3%82%8C%E3%81%A6%E3%81%84%E3%81%AA%E3%81%84%E5%8F%AF%E8%83%BD%E6%80%A7%E3%81%8C%E3%81%82%E3%82%8A%E3%81%BE%E3%81%99_(Delphi))
 - [インライン変数宣言 (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/%E3%82%A4%E3%83%B3%E3%83%A9%E3%82%A4%E3%83%B3%E5%A4%89%E6%95%B0%E5%AE%A3%E8%A8%80)


### ・W1050 set 式で WideChar がバイト文字に変換されました (Delphi)
**原因:**
Unicode 文字の集合に対して **in** 演算子を使った場合に発生する。多くの場合、ANSI 版 Delphi から Unicode 版 Delphi へのマイグレーション過程で発生。ANSI 版 Delphi 用しかないコンポーネントやライブラリを Unicode 版 Delphi で使おうとした時にも発生する。

**対処方法:**
殆どのケースでは `CharInSet()` で置き換えればいいだけだが、

```pascal
  for i:=1 to Length(S) do
//  if (S[1] in ['A'..'Z']) then
    if CharInSet(S[1], ['A'..'Z']) then // CharInSet() で置き換え
      ShowMessage('Hit!');
```

置換するとおかしくなりそうなロジックが含まれる場合には、**ワーニングをエラーに昇格**してエラーを徹底的に潰した方がいい。中途半端に型キャストで逃げると、後で問題が発生した場合に修正箇所を探すのが困難になる。

`W1050` が発生するユニットの先頭に次のコードを挿入すると、ワーニングがコンパイルエラーで停止するようになる。

```pascal
// "set 式で WideChar がバイト文字に変換されました(W1050)"をエラーに昇格
{$WARN WIDECHAR_REDUCED ERROR}
```

エラーをすべて潰さないと実行形式ファイルが作られないため、懸念される箇所を放置できなくなる。エラーをすべて解消してから当該コードを外すようにする。

**See also:**

 - [W1050 set 式で WideChar がバイト文字に変換されました (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1050_set_%E5%BC%8F%E3%81%A7_WideChar_%E3%81%8C%E3%83%90%E3%82%A4%E3%83%88_char_%E3%81%AB%E7%B8%AE%E5%B0%8F%E3%81%95%E3%82%8C%E3%81%BE%E3%81%97%E3%81%9F%EF%BC%88Delphi%EF%BC%89)
 - [System.SysUtils.CharInSet (DocWiki)](http://docwiki.embarcadero.com/Libraries/ja/System.SysUtils.CharInSet)
 - [<8> 集合型 (標準 Pascal 範囲内での Delphi 入門) (Qiita)](https://qiita.com/ht_deko/items/50bac0de384d66208944)

### ・W1055 Published によって型に RTTI ($M+) が追加されます '%s'
**原因:**
多くはコンポーネントでない (`TPersistent` の下位クラスでない) クラスのプロパティが **published** で公開されている事により発生する。

既存のコンポーネントクラスのロジックを~~パクって~~インスパイアしたクラスを作ったか、RTTI (実行時型情報) を使うにもかかわらず、コンパイラ指令を付け忘れているかのいずれかだと推察される。

**対処方法1:**
**published** にあるプロパティを **public** に移動する。基本はコレでいいと思う。

```pascal
type
  THoge = class(TObject)
    ...
  public
    ...
  published 
    property  FileName: string read FFileName write SetFileName;
  end;
```
↓

```pascal
type
  THoge = class(TObject)
    ...
  public
    ...
    property  FileName: string read FFileName write SetFileName;
  end;
```

**対処方法2:**
クラスを `TPersistent` から派生させる。

```pascal
type
  THoge = class(TPersistent)
```
**対処方法3:**
このワーニングを出しているクラスの宣言部を `{$TYPEINFO ON}` `{$TYPEINFO OFF}` (または `{$M+}` `{$M-}`) で括る。

```pascal
{$TYPEINFO ON}
  THoge = class(TObject)
    ...
{$TYPEINFO OFF}
```

**See also:**

 - [W1055 Published によって型に RTTI ($M+) が追加されます '%s' (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1055_Published_%E3%81%AB%E3%82%88%E3%81%A3%E3%81%A6%E5%9E%8B%E3%81%AB_RTTI_($M%2B)_%E3%81%8C%E8%BF%BD%E5%8A%A0%E3%81%95%E3%82%8C%E3%81%BE%E3%81%99_%27%25s%27_(Delphi))
 - [実行時型情報（Delphi）(DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/%E5%AE%9F%E8%A1%8C%E6%99%82%E5%9E%8B%E6%83%85%E5%A0%B1%EF%BC%88Delphi%EF%BC%89)
 - [Private 宣言、protected 宣言、public 宣言、published 宣言 (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/Private_%E5%AE%A3%E8%A8%80%E3%80%81protected_%E5%AE%A3%E8%A8%80%E3%80%81public_%E5%AE%A3%E8%A8%80%E3%80%81published_%E5%AE%A3%E8%A8%80)

### ・W1057 文字列の暗黙的なキャスト ('%s' から '%s') (Delphi)<br>・W1058 データ損失の可能性がある文字列の暗黙的なキャスト ('%s' から '%s') (Delphi)
**原因:**
異なる文字列型への代入で起こる。多くの場合、ANSI 版 Delphi から Unicode 版 Delphi へのマイグレーション過程で発生。ANSI 版 Delphi 用しかないコンポーネントやライブラリを Unicode 版 Delphi で使おうとした時にも発生する。

**対処方法:**
殆どのケースでは明示的に型キャストすればいいだけだが、型キャストするとおかしくなりそうなロジックが含まれる場合には、**ワーニングをエラーに昇格**してエラーを徹底的に潰した方がいい。中途半端に型キャストで逃げると、後で問題が発生した場合に修正箇所を探すのが困難になる。

`W1057` と `W1058` が発生するユニットの先頭に次のコードを挿入すると、ワーニングがコンパイルエラーで停止するようになる。

```pascal
// "文字列の暗黙的なキャスト(W1057)"をエラーに昇格
{$WARN IMPLICIT_STRING_CAST ERROR}

// "データ損失の可能性がある文字列の暗黙的なキャスト(W1058)"をエラーに昇格
{$WARN IMPLICIT_STRING_CAST_LOSS ERROR}
```

エラーをすべて潰さないと実行形式ファイルが作られないため、懸念される箇所を放置できなくなる。エラーをすべて解消してから当該コードを外すようにする。

**キャストで "逃げた" 場合:**
AnsiString を UnicodeString にキャストしても大きな問題はまず起きないが、UnicodeString を AnsiString にキャストすると文字の損失が起こる可能性が高い。

明示的にキャストしているコードのためのワーニングである `W1059` / `W1060` も用意されているが、デフォルトで無効になっている。キャストで逃げた箇所を再調査したい場合には、プロジェクトオプションでこの 2 つのワーニングを有効にする事ができる。

**See also:**

 - [W1057 文字列の暗黙的なキャスト ('%s' から '%s') (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1057_%E6%96%87%E5%AD%97%E5%88%97%E3%81%AE%E6%9A%97%E9%BB%99%E7%9A%84%E3%81%AA%E3%82%AD%E3%83%A3%E3%82%B9%E3%83%88_%28%27%25s%27_%E3%81%8B%E3%82%89_%27%25s%27%29_%28Delphi%29)
 - [W1058 データ損失の可能性がある文字列の暗黙的なキャスト ('%s' から '%s') (Delphi)](http://docwiki.embarcadero.com/RADStudio/ja/W1058_%E3%83%87%E3%83%BC%E3%82%BF%E6%90%8D%E5%A4%B1%E3%81%AE%E5%8F%AF%E8%83%BD%E6%80%A7%E3%81%8C%E3%81%82%E3%82%8B%E6%96%87%E5%AD%97%E5%88%97%E3%81%AE%E6%9A%97%E9%BB%99%E7%9A%84%E3%81%AA%E3%82%AD%E3%83%A3%E3%82%B9%E3%83%88_%28%27%25s%27_%E3%81%8B%E3%82%89_%27%25s%27%29_%28Delphi%29)
 - [W1059 文字列の明示的なキャスト ('%s' から '%s') (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1059_%E6%96%87%E5%AD%97%E5%88%97%E3%81%AE%E6%98%8E%E7%A4%BA%E7%9A%84%E3%81%AA%E3%82%AD%E3%83%A3%E3%82%B9%E3%83%88_(%27%25s%27_%E3%81%8B%E3%82%89_%27%25s%27)_(Delphi))
 - [W1060 データ損失の可能性がある文字列の明示的なキャスト ('%s' から '%s') (Delphi) (DocWiki)](http://docwiki.embarcadero.com/RADStudio/ja/W1060_%E3%83%87%E3%83%BC%E3%82%BF%E6%90%8D%E5%A4%B1%E3%81%AE%E5%8F%AF%E8%83%BD%E6%80%A7%E3%81%8C%E3%81%82%E3%82%8B%E6%96%87%E5%AD%97%E5%88%97%E3%81%AE%E6%98%8E%E7%A4%BA%E7%9A%84%E3%81%AA%E3%82%AD%E3%83%A3%E3%82%B9%E3%83%88_(%27%25s%27_%E3%81%8B%E3%82%89_%27%25s%27)_(Delphi))

# おわりに
とりあえず思いついたヒントとワーニングへの対処方法を並べてみました。
他にも何かあったら備忘録的に順次追記したいと思います。

## そういえば
Delphi の古い書籍や記事では、

 - 「IDE のコンパイラ指令に依存しないように、コードにコンパイラ指令を書いておけ！」
 - 「〔Ctrl〕+〔O〕,〔O〕でユニットの先頭に現在のコンパイラ指令が出力されるぞ！」

って、書いてあったりしますが、この記事を書いた時点の最新版 Delphi である 10.4 Sydney でそれをやると...

```Pascal
{$A8,B-,C+,D+,E-,F-,G+,H+,I+,J-,K-,L+,M-,N-,O+,P+,Q-,R-,S-,T-,U-,V+,W-,X+,Y+,Z1}
{$MINSTACKSIZE $00004000}
{$MAXSTACKSIZE $00100000}
{$IMAGEBASE $00400000}
{$APPTYPE GUI}
{$WARN SYMBOL_DEPRECATED ON}
{$WARN SYMBOL_LIBRARY ON}
{$WARN SYMBOL_PLATFORM ON}
{$WARN SYMBOL_EXPERIMENTAL ON}
{$WARN UNIT_LIBRARY ON}
{$WARN UNIT_PLATFORM ON}
{$WARN UNIT_DEPRECATED ON}
{$WARN UNIT_EXPERIMENTAL ON}
{$WARN HRESULT_COMPAT ON}
{$WARN HIDING_MEMBER ON}
{$WARN HIDDEN_VIRTUAL ON}
{$WARN GARBAGE ON}
{$WARN BOUNDS_ERROR ON}
{$WARN ZERO_NIL_COMPAT ON}
{$WARN STRING_CONST_TRUNCED ON}
{$WARN FOR_LOOP_VAR_VARPAR ON}
{$WARN TYPED_CONST_VARPAR ON}
{$WARN ASG_TO_TYPED_CONST ON}
{$WARN CASE_LABEL_RANGE ON}
{$WARN FOR_VARIABLE ON}
{$WARN CONSTRUCTING_ABSTRACT ON}
{$WARN COMPARISON_FALSE ON}
{$WARN COMPARISON_TRUE ON}
{$WARN COMPARING_SIGNED_UNSIGNED ON}
{$WARN COMBINING_SIGNED_UNSIGNED ON}
{$WARN UNSUPPORTED_CONSTRUCT ON}
{$WARN FILE_OPEN ON}
{$WARN FILE_OPEN_UNITSRC ON}
{$WARN BAD_GLOBAL_SYMBOL ON}
{$WARN DUPLICATE_CTOR_DTOR ON}
{$WARN INVALID_DIRECTIVE ON}
{$WARN PACKAGE_NO_LINK ON}
{$WARN PACKAGED_THREADVAR ON}
{$WARN IMPLICIT_IMPORT ON}
{$WARN HPPEMIT_IGNORED ON}
{$WARN NO_RETVAL ON}
{$WARN USE_BEFORE_DEF ON}
{$WARN FOR_LOOP_VAR_UNDEF ON}
{$WARN UNIT_NAME_MISMATCH ON}
{$WARN NO_CFG_FILE_FOUND ON}
{$WARN IMPLICIT_VARIANTS ON}
{$WARN UNICODE_TO_LOCALE ON}
{$WARN LOCALE_TO_UNICODE ON}
{$WARN IMAGEBASE_MULTIPLE ON}
{$WARN SUSPICIOUS_TYPECAST ON}
{$WARN PRIVATE_PROPACCESSOR ON}
{$WARN UNSAFE_TYPE OFF}
{$WARN UNSAFE_CODE OFF}
{$WARN UNSAFE_CAST OFF}
{$WARN OPTION_TRUNCATED ON}
{$WARN WIDECHAR_REDUCED ON}
{$WARN DUPLICATES_IGNORED ON}
{$WARN UNIT_INIT_SEQ ON}
{$WARN LOCAL_PINVOKE ON}
{$WARN MESSAGE_DIRECTIVE ON}
{$WARN TYPEINFO_IMPLICITLY_ADDED ON}
{$WARN RLINK_WARNING ON}
{$WARN IMPLICIT_STRING_CAST ON}
{$WARN IMPLICIT_STRING_CAST_LOSS ON}
{$WARN EXPLICIT_STRING_CAST OFF}
{$WARN EXPLICIT_STRING_CAST_LOSS OFF}
{$WARN CVT_WCHAR_TO_ACHAR ON}
{$WARN CVT_NARROWING_STRING_LOST ON}
{$WARN CVT_ACHAR_TO_WCHAR ON}
{$WARN CVT_WIDENING_STRING_LOST ON}
{$WARN NON_PORTABLE_TYPECAST ON}
{$WARN XML_WHITESPACE_NOT_ALLOWED ON}
{$WARN XML_UNKNOWN_ENTITY ON}
{$WARN XML_INVALID_NAME_START ON}
{$WARN XML_INVALID_NAME ON}
{$WARN XML_EXPECTED_CHARACTER ON}
{$WARN XML_CREF_NO_RESOLVE ON}
{$WARN XML_NO_PARM ON}
{$WARN XML_NO_MATCHING_PARM ON}
{$WARN IMMUTABLE_STRINGS OFF}
```

80 行のコンパイラ指令がユニットの先頭に追加されます。そのままだとかなりウザいです。`{$REGION}` `{$ENDREGION}` で括れない (畳めない) のが地味に痛い。インクルードファイル (*.inc) で処理すべきですかねぇ...。

```pascal:CompilerDirectives.inc
{$A8,B-,C+,D+,E-,F-,G+,H+,I+,J-,K-,L+,M-,N-,O+,P+,Q-,R-,S-,T-,U-,V+,W-,X+,Y+,Z1}
{$MINSTACKSIZE $00004000}
{$MAXSTACKSIZE $00100000}
...
{$WARN IMMUTABLE_STRINGS OFF}
```

```pascal:Unit1.pas
{$I 'CompilerDirectives.inc'}
unit Unit1;

interface
...
```